On a four-dimensional oncolytic Virotherapy model (ϵ=1)

This Mathematica Notebook is a supplementary material to the paper “On a three-dimensional and two four-dimensional
oncolytic viro-therapy models”. It contains some of the calculations and illustrations appearing in the paper.

Ep1)Section 3.5(in paper): 4-Dim.Viro-therapy model when ϵ=1

Ep1-1)Definition of the model and fixed points when ϵ=1

```
In[147]:= SetDirectory[NotebookDirectory[]];
AppendTo[$Path, Directory];
Clear["Global`*"];
Clear["K"];
(*Some aliases*)
Format[βy] = Subscript[β, y];
Format[βv] = Subscript[β, v];
Format[βz] = Subscript[β, z];


Unprotect[Power];
Power[0, 0] = 1;
Protect[Power];

par = {b, β, λ, δ, βy, βv, βz, c, γ, K, ϵ};
cp = Join[Thread[Drop[par, {1}] > 0], {b > 1}];
cKga1 = {K → 1, γ → 1};
cep1 = {ϵ → 1};
R0 = b β K / (β K + δ) (* Reproduction number*);

(*cnb={b→50};
cE1ri=Join[{βy→1/48,K→2139.258, β→.0002,λ→.2062,
   γ→1/18,δ→.025, βv→2*10^(-8),c→10^(-3),βz→.027},cep1];*)
cF1 = {β → 87/2, λ → 1, γ → 1/128, δ → 1 / 2, βy → 1, βv → 1, K → 1, βz → 1, c → 1, ϵ → 1};

(****** Four dim Deterministic  epidemic model with Logistic growth ****)
x1 = λ x (1 - (x + y) / K) -  β x v ;
y1 = β x v  - βy y z - γ y;
v1 = -β x v  - βv  v z + b γ y  - δ v;
z1 = z (βz  y  - c z^ϵ);
dyn = {x1, y1, v1, z1};
dyn3 = {x1, y1, v1} /. z → 0;(*3dim case used for E* *)
Print["  (x' y' v' z') =", dyn // FullSimplify // MatrixForm]
Print["b0=", b0 = b /. Apart[Solve[R0 == 1, b][[1]] // FullSimplify]]

(***Fixed point when z→0**)
eq = Thread[dyn3 == {0, 0, 0}];
sol = Solve[eq, {x, y, v}] // FullSimplify;
Es = {x, y, v} /. sol[[3]]; (*Endemic point with z=0*);
Print[" Endemic point with z=0 is E*=", Es // FullSimplify]
jacD = Grad[dyn /. cep1, {x, y, v, z}];
Print["J(x,y,v,z)="]
jacD // MatrixForm
Print["Det(J(x,y,v,z))="]
Det[jacD] // FullSimplify
```

$$\begin{pmatrix} x' \\ y' \\ v' \\ z' \end{pmatrix} = \begin{pmatrix} -v\,x\,\beta + x\left(1 - \frac{x+y}{K}\right)\lambda \\ v\,x\,\beta - y\left(z\,\beta_y + \gamma\right) \\ b\,y\,\gamma - v\,(x\,\beta + z\,\beta_v + \delta) \\ z\,(-c\,z^{\epsilon} + y\,\beta_z) \end{pmatrix}$$

$$b0 = 1 + \frac{\delta}{K\,\beta}$$

Endemic point with z=0 is E*=

$$\left\{\frac{\delta}{(-1+b)\,\beta},\ \frac{((-1+b)\,K\,\beta - \delta)\,\delta\,\lambda}{(-1+b)\,\beta\,((-1+b)\,K\,\beta\,\gamma + \delta\,\lambda)},\ \frac{\gamma\,((-1+b)\,K\,\beta - \delta)\,\lambda}{\beta\,((-1+b)\,K\,\beta\,\gamma + \delta\,\lambda)}\right\}$$

J(x,y,v,z) =

Out[175]//MatrixForm=

$$\begin{pmatrix} -v\,\beta - \frac{x\,\lambda}{K} + \left(1 - \frac{x+y}{K}\right)\lambda & -\frac{x\,\lambda}{K} & -x\,\beta & 0 \\ v\,\beta & -z\,\beta_y - \gamma & x\,\beta & -y\,\beta_y \\ -v\,\beta & b\,\gamma & -x\,\beta - z\,\beta_v - \delta & -v\,\beta_v \\ 0 & z\,\beta_z & 0 & -2\,c\,z + y\,\beta_z \end{pmatrix}$$

Det(J(x,y,v,z)) =

Out[177]=

$$\frac{1}{K}\Big(2\,c\,z\,\Big(K\,v\,\beta\,\left(z\,\beta_y + \gamma\right)\,(z\,\beta_v + \delta) + v\,x\,\beta\,(z\,\beta_v + \delta)\,\lambda - K\,\Big(x\,\beta\,\left(z\,\beta_y + \gamma - b\,\gamma\right) + \left(z\,\beta_y + \gamma\right)\,(z\,\beta_v + \delta)\Big)\,\lambda + (2\,x + y)\,\Big(x\,\beta\,\left(z\,\beta_y + \gamma - b\,\gamma\right) + \left(z\,\beta_y + \gamma\right)\,(z\,\beta_v + \delta)\Big)\,\lambda\Big) - \beta_z\,(K\,y\,\gamma\,((-1+b)\,x\,\beta - z\,\beta_v - \delta)\,\lambda - y\,(2\,x + y)\,\gamma\,((-1+b)\,x\,\beta - z\,\beta_v - \delta)\,\lambda + v\,x\,\beta\,(-2\,x\,z\,\beta_v + y\,\delta)\,\lambda + K\,v\,\beta\,(y\,\gamma\,(z\,\beta_v + \delta) + x\,z\,\beta_v\,\lambda))\Big)$$

In[178]:=

```
(*****Fixed points of 4-dim model using P(y)***)
fy=(c γ(b-1)-y βy βz);
gy=( βv βz y+c δ); hy=(γ +y βz βy/c);
xey=hy gy/(β fy); vey=y fy/gy; zey= βz y /c;
ys=y/.sol⟦3⟧(* y of E*     ****);

Py=λ(1-y/K)-β y fy/gy-λ hy gy/(β K fy); yb=c γ (b-1)/(βy βz);
Qy=λ fy gy(1-  y /K)- λ hy gy^2/(β K)-y β fy^2;
Qycol=Collect[Together[Qy],y];
Qycoef=CoefficientList[Qycol,y];
Print["f(y)=", fy, " ,g(y)=", gy, " , h(y)=", hy]
Print["p(y)=", Py//FullSimplify]
Print["The polynomial  Q(y) is of order  ", Length[Qycoef]-1]
Print["Coefficients of Q(y) are ", Qycoef//FullSimplify]

Dis=Collect[Discriminant[Qy,y],b];
Discoef=CoefficientList[Dis,b];Length[Discoef];
Disn=Dis//.cF1//N;
bL=50;
Plot[{0,Disn},{b,0,2 bL},AxesLabel→{"b","Dis"},PlotRange→{{0,bL},{-10,12}}]
Print["Roots of Dis[b]=0 are: ",solbE=Solve[Disn==0,b]]
```

$f(y) = -y\,\beta_y\,\beta_z + (-1+b)\,c\,\gamma$,$g(y) = y\,\beta_v\,\beta_z + c\,\delta$, $h(y) = \dfrac{y\,\beta_y\,\beta_z}{c} + \gamma$

$$p(y) = \frac{y\,\beta\,\left(y\,\beta_y\,\beta_z - (-1+b)\,c\,\gamma\right)}{y\,\beta_v\,\beta_z + c\,\delta} + \lambda - \frac{y\,\lambda}{K} - \frac{\left(\frac{y\,\beta_y\,\beta_z}{c} + \gamma\right)\,(y\,\beta_v\,\beta_z + c\,\delta)\,\lambda}{K\,\beta\,\left(-y\,\beta_y\,\beta_z + (-1+b)\,c\,\gamma\right)}$$

The polynomial $Q(y)$ is of order 3

Coefficients of $Q(y)$ are

$$\Big\{\frac{c^2\,\gamma\,((-1+b)\,K\,\beta - \delta)\,\delta\,\lambda}{K\,\beta},$$
$$-\frac{c\,\left(\beta_z\,\left(\delta\,\left(2\,\beta_v\,\gamma + \beta_y\,\delta\right) + K\,\beta\,\left(\beta_v\,\gamma - b\,\beta_v\,\gamma + \beta_y\,\delta\right)\right)\,\lambda + (-1+b)\,c\,\beta\,\gamma\,((-1+b)\,K\,\beta\,\gamma + \delta\,\lambda)\right)}{K\,\beta},$$
$$\frac{\beta_z\,\left(-\beta_v\,\beta_z\,\left(K\,\beta\,\beta_y + \beta_v\,\gamma + 2\,\beta_y\,\delta\right)\,\lambda + c\,\beta\,\left(2\times(-1+b)\,K\,\beta\,\beta_y\,\gamma + \beta_v\,(\gamma - b\,\gamma)\,\lambda + \beta_y\,\delta\,\lambda\right)\right)}{K\,\beta},$$
$$-\frac{\beta_y\,\beta_z^2\,\left(\beta_v^2\,\beta_z\,\lambda + c\,\beta\,\left(K\,\beta\,\beta_y - \beta_v\,\lambda\right)\right)}{c\,K\,\beta}\Big\}$$

Out[194]=



Roots of Dis[b]=0 are:

{{b → -126.518}, {b → -63.}, {b → -63.}, {b → -24.5518}, {b → 29.361}, {b → 45.9232}}

In[196]:=

```
QR=Solve[Qy==0,y,Cubics→False]//ToRadicals(*casus irreducibilis*);
ym= y/.QR[[2]];
yp=  y/.QR[[1]];yi=  y/.QR[[3]];
Em1={xey,ym,vey,zey}//.y→ym;
Ep1={xey,yp,vey,zey}//.y→yp;
Eim1={xey,yi,vey,zey}//.y→yi;
Es1=Join[{x,y,v}/.sol[[3]],{0}];
{Chop[yp],Chop[ym]}//.cF1//N;

jacE1K=jacD/.x→K/.y→0/.v→0/.z→0;
Print["J(EK)=", jacE1K//MatrixForm]
(*Jacobians of the fixed points**)
jEs1=jacD/.sol[[3]]/.z→0;
jEm1=jacD/.x→xey/.v→vey/.z→zey/.y→ym;
jEim1=jacD/.x→xey/.v→vey/.z→zey/.y→yi;
jEp1=jacD/.x→xey/.v→vey/.z→zey/.y→yp;
Print["Eig.val of J(EK) are:", Eigenvalues[jacE1K]//FullSimplify]
jacE1=jacD/.x→xey/.v→vey/.z→zey/.y→y//FullSimplify;
jacE1 //MatrixForm;
Det[jacE1]//FullSimplify;
Print[" Trace of either Ei, E+ or E- is : ", Tr[jacE1]//FullSimplify]
Print["J(E_*) is"]
jEs1//FullSimplify//MatrixForm
bbs=b/.Solve[yb==(y/.sol[[3]]),b][[1]](*long expression*);
```

$$\mathrm{J(EK)} = \begin{pmatrix} -\lambda & -\lambda & -K\beta & 0 \\ 0 & -\gamma & K\beta & 0 \\ 0 & b\gamma & -K\beta-\delta & 0 \\ 0 & 0 & 0 & 0 \end{pmatrix}$$

Eig.val of J(EK) are: $\left\{0, \frac{1}{2}\left(-K\beta-\gamma-\delta-\sqrt{-4\gamma\,(K\,(\beta-b\,\beta)+\delta)+(K\beta+\gamma+\delta)^2}\right), \frac{1}{2}\left(-K\beta-\gamma-\delta+\sqrt{-4\gamma\,(K\,(\beta-b\,\beta)+\delta)+(K\beta+\gamma+\delta)^2}\right), -\lambda\right\}$

Trace of either Ei, E+ or E- is : $-y\,\beta_z-\frac{y\,\beta_y\,\beta_z+c\,\gamma}{c}+\frac{y^2\,\beta\,\beta_y\,\beta_z}{y\,\beta_v\,\beta_z+c\,\delta}-\frac{(-1+b)\,c\,y\,\beta\,\gamma}{y\,\beta_v\,\beta_z+c\,\delta}+\frac{b\,\gamma\,(y\,\beta_v\,\beta_z+c\,\delta)}{y\,\beta_y\,\beta_z-(-1+b)\,c\,\gamma}+\lambda-\frac{y\,\lambda}{K}-\frac{2\left(\frac{y\,\beta_y\,\beta_z}{c}+\gamma\right)(y\,\beta_v\,\beta_z+c\,\delta)\,\lambda}{K\,\beta\left(-y\,\beta_y\,\beta_z+(-1+b)\,c\,\gamma\right)}$

J(E_*) is

Out[216]//MatrixForm=

$$\begin{pmatrix} \frac{\delta\lambda}{K\beta-bK\beta} & \frac{\delta\lambda}{K\beta-bK\beta} & -\frac{\delta}{-1+b} & 0 \\ \frac{\gamma\,((-1+b)\,K\beta-\delta)\,\lambda}{(-1+b)\,K\beta\gamma+\delta\lambda} & -\gamma & \frac{\delta}{-1+b} & \frac{\beta_y\,\delta\,(K\,(\beta-b\,\beta)+\delta)\,\lambda}{(-1+b)\,\beta\,((-1+b)\,K\beta\gamma+\delta\lambda)} \\ \frac{\gamma\,(K\,(\beta-b\,\beta)+\delta)\,\lambda}{(-1+b)\,K\beta\gamma+\delta\lambda} & b\gamma & \frac{b\,\delta}{1-b} & \frac{\beta_v\,\gamma\,(K\,(\beta-b\,\beta)+\delta)\,\lambda}{\beta\,((-1+b)\,K\beta\gamma+\delta\lambda)} \\ 0 & 0 & 0 & \frac{\beta_z\,((-1+b)\,K\beta-\delta)\,\delta\lambda}{(-1+b)\,\beta\,((-1+b)\,K\beta\gamma+\delta\lambda)} \end{pmatrix}$$

Ep1-2)Trace, Det and third criterion of Routh Hurwitz applied to E*:

Det and Trace of of E* and Analysis of the stability of E* in 4 dim when ϵ=1:

In[218]:=
```
Print["Tr[J[E*]]="]
trEs1=Tr[jacD//.Join[sol⟦3⟧,{z→0}]]//FullSimplify
Print["Det[J[E*]]="]
detEs1=Det[jacD//.Join[sol⟦3⟧,{z→0}]]//FullSimplify

pc=Collect[Det[ψ IdentityMatrix[4]-(jEs1)],ψ];
coT=CoefficientList[pc,ψ]//FullSimplify;
Length[coT]
a1=coT⟦4⟧//FullSimplify ;a2=coT⟦3⟧//FullSimplify;a3=coT⟦2⟧//FullSimplify ; a4=coT⟦1⟧//FullSi
Print["a1=",a1, ", a2=",a2, ", a3=",a3, " ,a4=", a4 ]
H4=a1*a2*a3-a3^2+a1^2 a4;
Print["H2(b0)=",H4/.b→b0//FullSimplify]
Print["Denominator of H2 is ",Denominator[Together[H4]]//FullSimplify]
ϕb4=Collect[Numerator[Together[H4]],b];
cofi=CoefficientList[ϕb4,b];
Length[cofi]
(*Print["value of ϕ(b) at crit b is "]
ϕb4/.b→b0//FullSimplify; (*so long expression*)*)
```

Tr[J[E*]]=

Out[219]= $$-\frac{K\,\delta\,(2\times(-1+b)\,\beta\,\gamma+b\,\beta\,\delta+\beta_z\,\delta)\,\lambda+\delta^2\,\lambda^2+(-1+b)\,K^2\,\beta\,(\beta\,\gamma\,((-1+b)\,\gamma+b\,\delta)-\beta_z\,\delta\,\lambda)}{(-1+b)\,K\,\beta\,((-1+b)\,K\,\beta\,\gamma+\delta\,\lambda)}$$

Det[J[E*]]=

Out[221]= $$-\frac{\beta_z\,\gamma\,\delta^2\,(K\,(\beta-b\,\beta)+\delta)^2\,\lambda^2}{(-1+b)^2\,K\,\beta^2\,((-1+b)\,K\,\beta\,\gamma+\delta\,\lambda)}$$

Out[224]= 5

$$a_1=\frac{K\,\delta\,(2\times(-1+b)\,\beta\,\gamma+b\,\beta\,\delta+\beta_z\,\delta)\,\lambda+\delta^2\,\lambda^2+(-1+b)\,K^2\,\beta\,(\beta\,\gamma\,((-1+b)\,\gamma+b\,\delta)-\beta_z\,\delta\,\lambda)}{(-1+b)\,K\,\beta\,((-1+b)\,K\,\beta\,\gamma+\delta\,\lambda)}$$

$$, a_2=\Big(\Big(\delta\,\lambda\,\Big(-\Big((-1+b)\,K^2\,\beta^2\,((-1+b)\,(\beta+\beta_z)\,\gamma+b\,\beta_z\,\delta)\Big)+(b\,\beta+\beta_z)\,\delta^2\,\lambda+ K\,\beta\,(\beta_z\,\delta\,((-1+b)\,\gamma+b\,\delta+\lambda-b\,\lambda)+(-1+b)\,\beta\,\gamma\,((-1+b)\,\gamma+\delta-\lambda+b\,(\delta+\lambda)))\Big)\Big)\Big/ \Big((-1+b)^2\,K\,\beta^2\,((-1+b)\,K\,\beta\,\gamma+\delta\,\lambda)\Big)\Big),\ a_3=$$

$$\Big(\Big(((-1+b)\,K\,\beta-\delta)\,\delta\,\lambda\,\Big(\delta^2\,\Big((-1+b)^2\,\beta\,\gamma-b\,\beta_z\,\delta\Big)\,\lambda^2+(-1+b)^2\,K^2\,\beta^2\,\gamma\,\Big((-1+b)^2\,\beta\,\gamma^2+\beta_z\,\delta\,\lambda\Big)+ (-1+b)\,K\,\beta\,\gamma\,\delta\,\lambda\,\Big(2\,(-1+b)^2\,\beta\,\gamma-\beta_z\,((-1+b)\,\gamma+\delta-\lambda+b\,(\delta+\lambda))\Big)\Big)\Big)\Big/ \Big((-1+b)^3\,K\,\beta^2\,((-1+b)\,K\,\beta\,\gamma+\delta\,\lambda)^2\Big)\Big)\ ,a_4=-\frac{\beta_z\,\gamma\,\delta^2\,(K\,(\beta-b\,\beta)+\delta)^2\,\lambda^2}{(-1+b)^2\,K\,\beta^2\,((-1+b)\,K\,\beta\,\gamma+\delta\,\lambda)}$$

H2(b0)=0

Denominator of H2 is $(-1+b)^6\,K^3\,\beta^5\,((-1+b)\,K\,\beta\,\gamma+\delta\,\lambda)^4$

Out[232]= 11

Numerical values ϵ=1:

In[233]:=
```
cn=Join[cF1]; cnb={b→40};
cb=NSolve[(ϕb4//.cn)==0,b,WorkingPrecision→10]
bM=Max[Table[Re[b/.cb[[i]]],{i,Length[cb]}]];
Print["bH=",bH=N[bM,30]]
Print["b0=",b0/.cn//N]

Print["E*",Es1//.cn/.cnb//N]
Print["E+=",Em1//.cn/.cnb//N]
Print["Eim=",Eim1//.cn/.cnb//N]
Print["roots of Dis[b]=0:", bcE1=NSolve[(Dis//.cn)==0,b]]
bc1=Chop[Evaluate[b/.bcE1[[5]]]];
bc2=Chop[Evaluate[b/.bcE1[[6]]]];
Print["b1*=", bc1]
Print[" b2*=", bc2]
```

Out[234]= {{b → −258.7317419}, {b → −0.01615999598}, {b → 0.060848063 − 10.686990738 i}, {b → 0.060848063 + 10.686990738 i}, {b → 0.8448282668 − 0.9299250641 i}, {b → 0.8448282668 + 0.9299250641 i}, {b → 0.8878046288}, {b → 1.011494253}, {b → 1.022996712}, {b → 1.152364263}}

bH=1.152364263

b0=1.01149

E*{0.000294724, 0.0363426, 0.0221463, 0.}

E+= $\{0.00262408 + 5.51068\times10^{-19}\, i,\ 0.0462057 + 8.67362\times10^{-18}\, i,\ 0.021866 + 3.02367\times10^{-18}\, i,\ 0.0462057 + 8.67362\times10^{-18}\, i\}$

Eim= $\{0.114678 - 9.26732\times10^{-17}\, i,\ 0.263221 - 2.77556\times10^{-17}\, i,\ 0.0143012 + 8.58446\times10^{-18}\, i,\ 0.263221 - 2.77556\times10^{-17}\, i\}$

roots of Dis[b]=0:
{{b→-126.518}, {b→-63.}, {b→-63.}, {b→-24.5518}, {b→29.361}, {b→45.9232}}

b1*=29.361

b2*=45.9232

Ep1-3)Bifurcation diagrams :

Numerical solution of the stability (Bifurcation diagram) wrt y:

In[ ]:=
```
(*Checks on the stability of the fixed points**)
Print["Eigenvalues of E* when b=20 and when b=40, respectively "]
Eigenvalues[jEs1] //. cF1 /. b → 20 // N
Eigenvalues[jEs1] //. cF1 /. b → 40 // N
Print["Eigenvalues of E+  when b=40"]
Chop[Eigenvalues[jEp1 //. cF1 /. b → 20 // N]]

Print["Eigenvalues of Eim  when b=40"]
Chop[Eigenvalues[jEim1 //. cF1 /. b → 40 // N]]
Print["Eigenvalues of E- between b1* and b2* "]
Chop[Eigenvalues[jEm1 //. cF1 /. b → 35 // N]]
Print["Eigenvalues of E- between 0 and b1*"]
Chop[Eigenvalues[jEm1 //. cF1 /. b → 20 // N]]
```

Eigenvalues of E* when b=20 and when b=40, respectively

Out[ ]= {0.0718263, -0.586224, 0.0257451 - 0.0774375 i, 0.0257451 + 0.0774375 i}

Out[ ]= {0.0363426, -0.555066, 0.017069 - 0.082122 i, 0.017069 + 0.082122 i}

Eigenvalues of E+ when b=40

Out[ ]= {-36.8126, -0.800277, -0.073135 + 0.126319 i, -0.073135 - 0.126319 i}

Eigenvalues of Eim when b=40

Out[ ]= {-24.7847, -0.377551, -0.150826 + 0.260466 i, -0.150826 - 0.260466 i}

Eigenvalues of E- between b1* and b2*

Out[ ]= {-4.10357, 0.346228, -0.0664887 + 0.180685 i, -0.0664887 - 0.180685 i}

Eigenvalues of E- between 0 and b1*

Out[ ]= {-0.840551 + 1.00512 i, 0.242338 - 0.343832 i,
0.0933864 + 0.157103 i, -0.0937105 + 0.0144354 i}

In[ ]:=
```
cut=cF1;bL=50;max=0.37;
b0n=b0//.cn;
b1=Chop[bbs//.cF1//N];
Print["b0=",b0n//N, " ,b_b*=", b1, ", b1*=", bc1, " , b2*=", bc2]
lin1=Line[{{ bc1,0},{ bc1,max}}];
li1=Graphics[{Thick,Black,Dashed,lin1}];
lin2=Line[{{ bc2,0},{ bc2,max}}];
li2=Graphics[{Thick,Black,Dashed,lin2}];
lin3=Line[{{ b0n,0},{ b0n,max}}];
li3=Graphics[{Thick,Black,Dashed,lin3}];
lin4=Line[{{ b1,0},{ b1,max}}];
li4=Graphics[{Thick,Black,Dashed,lin4}];
pyb=Plot[{yb}//.cut,{b,0,bL},PlotStyle→{Dashed,Thick,Cyan},
PlotRange→{{0,200},{0,max}},PlotLegends→{"y_b "}];
(*pym1n=Plot[{ym}//.cut,{b,0,bL},PlotStyle→{Green,Thick},PlotRange→All,PlotPoints→200,
PlotLegends→{"E_ unstable"}];*)

pym=Plot[{ym}//.cut,{b,0,bL},PlotStyle→{Green,Dashed,Thick},PlotRange→All,PlotPoints→180,
PlotLegends→{"E_ unstable"}];
pyK1=Plot[0,{b,0,b0n},PlotStyle→{Magenta},PlotRange→All,PlotLegends→{"E_K stable"}];
pyK2=Plot[0,{b,b0n,bL},PlotStyle→{Blue},PlotRange→All,PlotLegends→{"E_K unstable"}];
pyi=Plot[{yi}//.cut,{b,0, bL},PlotStyle→{Red,Dashed,Thick},PlotRange→All,
PlotLegends→{"E_im unstable"}];
pyp=Plot[{yp}//.cut,{b,b0n,bL},PlotStyle→{Purple, Thick},PlotRange→All,
PlotLegends→{"E_+ stable"}];
pypn=Plot[{yp}//.cut,{b,0,b0n},PlotStyle→{Yellow, Thick},PlotRange→All,
PlotLegends→{"E_+ (y<0)"}];
N[{yp}//.cut/.b→40,20](*check*)
Print["Q(yp) at b0 is "]
Qy/.y→yp/.b→b0//FullSimplify
Show[pyp,pyb,li3,li1,li2,li4];
pys1=Plot[{ys}//.cut,{b,0,b1},PlotStyle→{Orange, Dotted},
PlotRange→{{0,200},{0,max}},
PlotLegends→{"E_* outside domain"}];
pys2=Plot[{ys}//.cut,{b,b1,bL},PlotStyle→{Orange,Thick,Dashed},
PlotRange→{{0,200},{0,max}},
PlotLegends→{"E_* unstable"}];


Print["y*'(b0)=",D[ys,b]/.b→b0n//.cut//N//FullSimplify]
Chop[ys/.b→b0n//.cut//N](*Check*);
bifep1=Show[{pys2,pys1,pyi,pyp,pypn,pyK1,pyK2,pym,pyb,li3,li1,li2,li4},
Epilog→{Text["b_0",Offset[{-2,10},{ b0n//.cut,0}]],{PointSize[Large],
Style[Point[{ b0n//.cut,0}],Black]},
Text["b_1*",Offset[{-8,10},{ bc1,0}]],{PointSize[Large],
Style[Point[{ bc1,0}],Purple]},
Text["b_2*",Offset[{10,10},{ bc2,0}]],{PointSize[Large],
Style[Point[{ bc2,0}],Yellow]},Text["b_b*",Offset[{10,10},{ b1,0}]],{PointSize[Large],
Style[Point[{ b1,0}],Magenta]}},AxesLabel→{"b","y"},PlotRange→{{-0.2,bL},{-0.1,max}}]
Export["EriB.pdf",bifep1]
```

b0=1.01149 ,b_{b*}=14.0011, b1*=29.361 , b2*=45.9232

Out[ ]= $\{0.29448140956728204616 + 0. \times 10^{-21}\, i\}$

Q(yp) at b0 is

Out[∘]= 0

y*'(b0)=86.3256

Out[∘]=



Out[∘]= EriB.pdf

(x,b)-Bifurcation diagram:

Determination of the endemic points with respect to x when ϵ=1:

In[∘]:= **Solve[((y1) /. vex /. y → yex) == 0, z]**

Out[∘]=
$$\left\{\left\{z \to \frac{-c\,\gamma - \frac{c\,x\,\lambda}{K} - \beta_z \sqrt{\frac{4\,c\,\beta_y \left(x\,\lambda - \frac{x^2\,\lambda}{K}\right)}{\beta_z} + \left(-\frac{c\,\gamma}{\beta_z} - \frac{c\,x\,\lambda}{K\,\beta_z}\right)^2}}{2\,c\,\beta_y}\right\},\ \left\{z \to \frac{-c\,\gamma - \frac{c\,x\,\lambda}{K} + \beta_z \sqrt{\frac{4\,c\,\beta_y \left(x\,\lambda - \frac{x^2\,\lambda}{K}\right)}{\beta_z} + \left(-\frac{c\,\gamma}{\beta_z} - \frac{c\,x\,\lambda}{K\,\beta_z}\right)^2}}{2\,c\,\beta_y}\right\}\right\}$$

In[640]:=
```
yex=c z/βz;(*Frome Solve[(z1/z/.cep1)==0,y]*)
Print["ye(x)=",yex]
Print["ve(x)="]
vex=Solve[(x1/x)==0,v][[1]]
Print["ze(x)="]
zex=Solve[((y1)/.vex/.y→yex)==0,z][[2]]//FullSimplify
(*the first solution of z above doesn't belong to the domain*)
(v1/.vex/.y→yex/.zex );
xex=Solve[((v1/.vex/.y→yex/.zex )/.cF1)==0,x,Cubics→False];
(*or // ComplexExpand[#, TargetFunctions → {Re, Im}] &*)
(*so long time when it's not numeric**)

Print["Number of endemic x"]
Length[xex]
Print["Numerical check"]
xex/.b→40//N
```

$\text{ye}(x) = \frac{c\, z}{\beta_z}$

ve(x) =

Out[643]= $\left\{v \to \frac{(K - x - y)\, \lambda}{K\, \beta}\right\}$

ze(x) =

Out[645]= $\left\{z \to \frac{-c\,(K\,\gamma + x\,\lambda) + K\,\beta_z\,\sqrt{\frac{c\,\left(4\,K\,(K-x)\,x\,\beta_y\,\beta_z\,\lambda + c\,(K\,\gamma + x\,\lambda)^2\right)}{K^2\,\beta_z^2}}}{2\,c\,K\,\beta_y}\right\}$

Solve: Solutions may not be valid for all values of parameters.

Number of endemic x

Out[649]= 4

Numerical check

Out[651]= {{x → 1.}, {x → 0.00262408}, {x → 0.114678}, {x → 0.540959}}

```
In[846]:= (*Jacobians of the fixed points**)
jEsx = jacD /. cep1 /. sol[[3]] /. z → 0;
jEmx = jacD /. cep1 /. vex /. y → yex /. zex /. xex[[2]];
jEimx = jacD /. cep1 /. vex /. y → yex /. zex /. xex[[3]];
jEpx = jacD /. cep1 /. vex /. y → yex /. zex /. xex[[4]];
(*Checks on the stability of the fixed points**)
Print["Eigenvalues of E* : "]
Print[" between b0 and b1* "]
Eigenvalues[jEsx] //. cF1 /. b → 15 // N
Print[" between b1* and b2* "]
Eigenvalues[jEsx] //. cF1 /. b → 40 // N
Print["Eigenvalues of E+  between b1* and b2*"]
Chop[Eigenvalues[jEpx //. cF1 /. b → 42 // N]]
Print["Eigenvalues of Eim  between b1* and b2*"]
Chop[Eigenvalues[jEimx //. cF1 /. b → 35 // N]]
Print["Eigenvalues of E- between b0 and b1* "]
Chop[Eigenvalues[jEmx //. cF1 /. b → 15 // N]]
Print["Eigenvalues of E-  between b1* and b2*"]
Chop[Eigenvalues[jEimx //. cF1 /. b → 40 // N]]
Print["Eigenvalues of E-  after b2*"]
Chop[Eigenvalues[jEimx //. cF1 /. b → 50 // N]]

Eigenvalues of E* :

 between b0 and b1*

Out[852]= {0.0950185, -0.606269, 0.0309604 - 0.074022 i, 0.0309604 + 0.074022 i}

 between b1* and b2*

Out[854]= {0.0363426, -0.555066, 0.017069 - 0.082122 i, 0.017069 + 0.082122 i}

Eigenvalues of E+  between b1* and b2*

Out[856]= {-22.81, -0.157506 + 0.275785 i, -0.157506 - 0.275785 i, -0.301641}

Eigenvalues of Eim  between b1* and b2*

Out[858]= {-4.10357, 0.346228, -0.0664887 + 0.180685 i, -0.0664887 - 0.180685 i}

Eigenvalues of E- between b0 and b1*

Out[860]= {-38.9444, -0.864225, -0.053898 + 0.0931984 i, -0.053898 - 0.0931984 i}

Eigenvalues of E-  between b1* and b2*

Out[862]= {-6.50192, 0.307571, -0.103153 + 0.233849 i, -0.103153 - 0.233849 i}

Eigenvalues of E-  after b2*

Out[864]= {-14.0734 + 7.77106 i, -0.136004 + 0.346012 i,
  -0.181515 - 0.313963 i, -0.00338735 + 0.318699 i}
```

In[1384]:=

```
cut=cF1;bL=49;max=1.1;
b0n=b0//.cn;
b1=Chop[b/.Solve[(Es1[[1]])==(1-yb),b][[2]]//.cF1//N]
Print["bH=",bH, " ,b1*=",bc1, " ,b2*=",bc2]
lin1=Line[{{ bc1,0},{ bc1,max}}];
li1=Graphics[{Thick,Black,Dashed,lin1}];
lin2=Line[{{ bc2,0},{ bc2,max}}];
li2=Graphics[{Thick,Black,Dashed,lin2}];
lin3=Line[{{ b0n,0},{ b0n,max}}];
li3=Graphics[{Thick,Black,Dashed,lin3}];
lin4=Line[{{ b1,0},{ b1,max}}];
li4=Graphics[{Thick,Black,Dashed,lin4}];
pxp=Plot[{x/.xex[[4]]}//.cut,{b,0,bL},PlotStyle→{Purple,Thick},PlotRange→All,PlotPoints→180,
PlotLegends→{"E+ stable"}]
pxK1=Plot[K//.cut,{b,0,b0n},PlotStyle→{Magenta},PlotRange→All,PlotLegends→{"EK stable"}];
pxK2=Plot[K//.cut,{b,b0n,bL},PlotStyle→{Blue},PlotRange→All,PlotLegends→{"EK unstable"}];
pxi=Plot[{x/.xex[[3]]}//.cut,{b,b0n, bL},PlotStyle→{Red, Dashed,Thick},PlotRange→All,
PlotLegends→{"Eim unstable"}]
pxm0=Plot[{x/.xex[[2]]}//.cut,{b,0,b0n},PlotStyle→{Yellow,  Thick},PlotRange→All,
PlotLegends→{"E- outside domain "}]
pxm=Plot[{x/.xex[[2]]}//.cut,{b,b0n,bc1},PlotStyle→{Green,  Thick},PlotRange→All,
PlotLegends→{"E- stable"}]
pxm1=Plot[{x/.xex[[2]]}//.cut,{b,bc2,bL},PlotStyle→{Green, Thick},PlotRange→All,PlotPoints→40]
pxm2=Plot[{x/.xex[[2]]}//.cut,{b,bc1,bc2},PlotStyle→{Green,Dashed,  Thick},PlotRange→All,
PlotLegends→{"E- unstable"}]

pxs=Plot[{x/.sol[[3]]}//.cut,{b,0,bL},PlotStyle→{Orange,Thick,Dashed},
PlotRange→{{0,bL},{0,max}},
PlotLegends→{"E* unstable"}]

pxb=Plot[{1-yb}//.cut,{b,0,bL},PlotStyle→{Dashed,Thick,Cyan},
PlotRange→{{0,bL},{0,max}},PlotLegends→{"1-yb "}];

bifep1=Show[{pxs,pxi,pxp,pxK1,pxK2,pxm,pxm1,pxm2,(*pxm0,*)pxb,li3,li1,li2,li4},
Epilog→{Text["b0",Offset[{10,10},{ b0n//.cut,0}]],{PointSize[Large],
Style[Point[{ b0n//.cut,0}],Black]},
Text["b1*",Offset[{-8,10},{ bc1,0}]],{PointSize[Large],
Style[Point[{ bc1,0}],Purple]},
Text["b2*",Offset[{10,10},{ bc2,0}]],{PointSize[Large],
Style[Point[{ bc2,0}],Yellow]},Text["bb*",Offset[{0,10},{ b1,0}]],{PointSize[Large],
Style[Point[{ b1,0}],Magenta]}},AxesLabel→{"b","x"},PlotRange→{{-0.1,bL},{-0.1,max}}]
Export["Bif1x.pdf",bifep1]
```

Out[1386]= 128.989

bH=1.152364263 ,b1*=29.361 ,b2*=45.9232

Out[1396]=



Out[1399]=



Out[1400]=



Out[1401]=



Out[1402]=



Out[1403]=



Out[1404]=



Out[1406]=



```
Out[1407]= Bif1x.pdf

In[1372]:= Print["between b0 and b1*; E-="]
           {x /. xex⟦2⟧, yex⟦2⟧ /. zex /. xex⟦2⟧,
               v /. vex /. y → yex /. (zex) /. xex⟦2⟧, z /. zex /. xex⟦2⟧} //. cut /. b → 15 // N
           Print["with yb="]
           yb //. cut /. b → 15 // N
           Print["between b1* and b2*; E-="]
           {x /. xex⟦2⟧, yex⟦2⟧ /. zex /. xex⟦2⟧,
               v /. vex /. y → yex /. (zex) /. xex⟦2⟧, z /. zex /. xex⟦2⟧} //. cut /. b → 35 // N
           Print["between after b2*; E-="]
           {x /. xex⟦2⟧, yex⟦2⟧ /. zex /. xex⟦2⟧,
               v /. vex /. y → yex /. (zex) /. xex⟦2⟧, z /. zex /. xex⟦2⟧} //. cut /. b → 47 // N
           Print["between between b0 and  b1*; E*="]
           sol⟦3⟧ //. cut /. b → 15 // N
           Print["between between b0 and  b1*; E*="]
           sol⟦3⟧ //. cut /. b → 35 // N

           between b0 and b1*; E-=

Out[1373]= {0.878336, 0.107541, 0.000324678, 0.107541}

           with yb=

Out[1375]= 0.109375

           between b1* and b2*; E-=

Out[1377]= {0.00405931, 0.0579238, 0.0215636, 0.0579238}

           between after b2*; E-=

Out[1379]= {0.0017057, 0.0367794, 0.0221038, 0.0367794}

           between between b0 and  b1*; E*=

Out[1381]= {x → 0.000821018, y → 0.0950185, v → 0.0207853}

           between between b0 and  b1*; E*=

Out[1383]= {x → 0.000338066, y → 0.0414636, v → 0.0220275}
```